
                           Работа со строками
   Строка -  это массив символов,  т.е.  элементов типа char.  В Паскале

строке соответствует тип String.

     VAR Имя_строки : string [Длина];

   Если длина не указана, выделяется память под строку до 255 символов.

   var s1:string;

       s2:string[20];

       s3:array [1..20] of string;

   Операции со строками:   

   Присвоить строке значение - оператор присваивания.

   s1:='А.И. Иванов';
   s1[3]:=’В’; {символу присвоили символ!}

   s2:='1999';

   s3[1]:='Информатика';

   s3[2]:='';

   Ввести строку: только оператор readln.

   Вывести строку: Write или Writeln

     s2:='SUMMA';

     Write (s2)   ;  - будет выведено  SUMMA

     Writeln ('Сумма':10);  - будет выведено    _ _ _ _ _ Сумма
   Соединить 2 строки - оператор сложения.

   s1:='1999' + ' год';

   s3[3]:=s1+' '+s2;

   Если при сложении строк превышена максимальная длина  строки,  лишние

символы отсекаются.

   Сравнить 2 строки: операции сравнения. Правила:

     - строки равны только при одинаковом наборе символов и одинаковой длине;

     - иначе происходит поэлементное сравнение  символов по их кодам:

       '0'<'1'<...<'9'<'A'<...<'Z'<'a'<...<'z'<символы кириллицы

   Стандартные подпрограммы обработки строк:
     Length (s:string):integer - определить длину строки

     Copy (str:string; N,L:integer): string - возвращает часть строки  str

длиной L, начиная с позиции N

     Insert (s0:string; var s: string; N: integer) - в строку s вставляет
строку s0, начиная с позиции N

     Delete (var s:string; N,L:integer) - в строке s удаляет L символов,

начиная с позиции N.

     Pos (s0,  s:string): integer - возвращает позицию, начиная с которой

строка s0 содержится в строке s или 0, если s0 не содержится в s

     Str (X: real; var s:string)

     Str (X: integer; var s:string) - преобразует число X в строку s.

     Val (s:string; var X: real; error:integer);

     Val (s:string; var X: integer; var error:integer); - преобразует
строку s в число X. Если преобразовать удалось, error=0, иначе error=

номеру первого непреобразуемого символа.

     Пример 1: разбираем предложение на слова и выводим каждое слово


var s,w:string;


    p:integer;


begin


 writeln ('Введите текст');


 readln (s);


 p:=1;


 while p<>0 do begin


  p:=pos (' ',s);


  if p=0 then w:=copy (s,1,length(s))


  else w:=copy (s,1,p-1);


  delete (s,1,p);


  writeln (w);


 end;


end.     

     Пример 2: удаляем лишние пробелы между словами


{ . . . }

 p:=1;


 while p<>0 do begin


  p:=pos ('  ',s);


  if p>0 then delete (s,p,1);


 end;


 if s[1]=' ' then delete (s,1,1);


 if s[length(s)]=' ' then delete (s, length(s),1);


 writeln (s);

Пример 3. Работаем со строкой как с массивом символов (считаем кол-во пробелов в строке)

var s:string;


 k,i:integer;


begin


 writeln ('Text?');


 readln (s);


 k:=0;


 for i:=1 to length (s) do


  if s[i]=' ' then k:=k+1;


 writeln ('k=',k);


end.

                            Работа с файлами
     1) Общие операции.

     Описать переменную типа text:


     var f:text;

     Связать ее с именем файла на диске:

     
assign (f,'имя_файла');

     Открыть файл для чтения или записи данных:

      reset (f) - открыть для чтения

      rewrite (f) - открыть для записи

     Имя файла может содержать путь MS-DOS: 'c:\student\data1.txt'

     Для завершения работы использовать процедуру

      close(f);

     2) Чтение и запись данных - стандартные процедуры read, readln, write,

writeln. Первым параметром указывается имя файловой переменной.
      Пример 1) Чтение числовых данных из файла
          var f,w:text;

          a: array [1..5] of integer;

          i:integer;

      begin

       assign (f,'data.txt');

       reset (f);

       for i:=1 to 5 do read (f, a[i]);

       close (f);

       assign (w, 'result.dat');

       rewrite (w);

       writeln (w,'Результаты:');

       for i:=1 to 5 do writeln (w,a[i]:5,sqr(a[i]):5);

       close(w);

      end.

      Файл data.txt может быть, например, таким:

      Данные:
      1 2 3
      4 5
      Файл result.dat будет таким:



Результаты:



    1    1



    2    4



    3    9



    4   16



    5   25

Проверки при работе с файлами:

      Eof (f) - истина, если достигнут конец файла

      Eoln (f) - истина, если достигнут конец строки

      Filesize (f) - возвращает размер файла в символах

      Filepos(f) - возвращает текущее положение чтения в файле

      Seek (f,N) - делает текущим символ N в файле f

      Пример 2. Просмотр текстового файла на экране:

var f:text;

    s:string;

begin
    write (‘File name? ’);
     readln (s);
     assign (f,s);

     reset (f);

     while not eof(f) do begin

           readln (f,s);

           writeln (s);

     end;

     close (f);
end.
Пример 3. Работаем со строковым файлом данных
Файл data.txt имеет следующий вид:
Иванов 2

Петров 1

Сидоров 3
Попов 2
(фамилия – номер участка)

Напишем программу для вывода фамилий всех работников выбранного участка и подсчета их количества. 

var f:text;

 s,fam:string;

 u,uch,p,kol,i:integer;

begin

 writeln ('Участок?');

 read (uch);

 assign (f,'data.txt');

 reset (f);

 kol:=0;

 while not eof (f) do begin

  readln (f,s);

  p:=pos(' ',s);

  if p>0 then begin

   fam:=copy (s,1,pos(' ',s));

   delete (s,1,pos(' ',s));

   val (s,u,i);

   if i<>0 then begin

    writeln ('Ошибка в числе ',s);

    halt;

   end;

   if u=uch then begin {подсчет рабочих на участке}
    writeln (fam);

    kol:=kol+1;

   end;

  end

  else begin

   writeln ('Ошибка в строке ',s);

   halt;

  end;

 end;

 close (f);

 writeln ('kol=',kol);

end.
Программу можно упростить, изменив формат исходных данных – например, на одной строке файла может быть набрана фамилия, а на другой – номер участка. Тогда основной цикл чтения файла будет выглядеть так:

while not eof (f) do begin
  readln (f,fam);

  readln (f,u);

  if (u=uch) then begin


{ . . . }

  end;
end;
